Measuring quantum radiation reaction in laser–electron-beam collisions

T G Blackburn 2015 Plasma Phys.Control.Fusion 57 075012

Notebook: Óscar Amaro, February 2021 @ GoLP-EPP
Contact: oscar.amaro@tecnico.ulisboa.pt

Introduction

In this notebook we reproduce Figure B1 of the article.

Physical constants

In[1]:=
```
(*α fine structure constant*)
α = 1 / 137; (*[#]*)
(*c speed of light in vacuum*)
c = 299 792 458; (*[m/s]*)
(*m electron mass*)
m = 0.5109989461 / 1000; (*[GeV]*)
(*ħ Planck's constant*)
ħ = 6.626070040 * 10 ^ - 34 / (2 π);
(*e electron abs charge*)
e = 1.6021766208 * 10 ^ - 19;
(*ϵ vaccum permitivity [F/m]*)
ϵ0 = 8.854 × 10 ^ - 12;
```

Figure 4

In[7]:=
```
(* Clear variables *)
Clear[μ, σ, ϵ, dNdϵ, rectArea, eff, distArea]
```

In[8]:=
```
(* Figure 4 electron beam *)
dNdϵ[ϵ_, μ_, σ_] := (μ + σ / 3 - ϵ) ^ (-3 / 2) Exp[- σ / (2 (μ + σ / 3 - ϵ))]
distArea[μ_, σ_] := NIntegrate[dNdϵ[ϵ, μ, σ], {ϵ, 0, μ + σ / 3}]

Manipulate[Plot[dNdϵ[ϵ, μ1, σ1], {ϵ, 0, μ1 + σ1 / 3},
  Frame → True, FrameLabel → {"ϵ", "dN/dϵ"}, PlotRange → All,
  PlotStyle → Directive[Black, Dashed], Axes → False, Frame → True,
  PlotLabel → "PDF of energy distribution"], {μ1, 100, 1000}, {σ1, 100, 1000}]
```

Out[10]=



In[11]:=
```
(* get maximum of distribution *)
res = D[dNdϵ[ϵ, μ, σ], ϵ] // Simplify;
Print["d/dϵ dN/dϵ =", res]
```

$$\frac{d}{d\epsilon}\frac{dN}{d\epsilon} = -\frac{81\sqrt{3}\, e^{\frac{3\sigma}{6\epsilon-6\mu-2\sigma}}\,(\epsilon-\mu)}{2\,(-3\epsilon+3\mu+\sigma)^{7/2}}$$

In[13]:=
```
(* get maximizing energy *)
Solve[D[dNdϵ[ϵ, μ, σ], ϵ] == 0, ϵ][[1, 1]]
```

Out[13]= $\epsilon \to \mu$

In[14]:=
```
(* distribution at peak *)
maxdNdϵ = dNdϵ[μ, μ, σ]
```

Out[14]= $\dfrac{3\sqrt{3}}{e^{3/2}\,\sigma^{3/2}}$

In[15]:=
```
(* limit of energy domain *)
maxϵ = μ + σ / 3;
μ = 400;
σ = 20;
LogPlot[dNdϵ[x, μ, σ], {x, 0.99 maxϵ, maxϵ}]
Clear[μ, σ]

Limit[dNdϵ[x, μ, σ], x → ∞]
```

Out[18]=



Out[20]= 0

In[21]:=
```
(* this distribution has ≠ 0 value at origin *)
dNdϵ[0, μ, σ]
```

Out[21]=

$$\frac{e^{-\frac{\sigma}{2\left(\mu+\frac{\sigma}{3}\right)}}}{\left(\mu+\frac{\sigma}{3}\right)^{3/2}}$$

Appendix A

In[22]:=
```
Clear[F, Fapprox, η, χ]
F[η_, χ_] := (4 χ)/(3 η^2) ((1 - (2 χ)/η + 1/(1 - 2 χ / η)) BesselK[2 / 3, (4 χ)/(3 η^2 (1 - 2 χ / η))] -
    NIntegrate[BesselK[1 / 3, t], {t, (4 χ)/(3 η^2 (1 - 2 χ / η)), ∞}])
Fapprox[η_, χ_] := (16 / 3) ^ (1 / 3) Gamma[2 / 3] η^ (-2 / 3) χ^ (1 / 3)
η = 1;
LogLogPlot[{F[η, χ], Fapprox[η, χ]}, {χ, 0.001 η, η / 2}, PlotPoints → 3,
 AxesLabel → {"χ", "F(η=1,χ)"}, PlotStyle → {Default, Dashed}]
Clear[
 η]
```

Out[26]=



Low χ expansion

In[28]:=
```
Clear[η, χ]
Refine[Series[(4 χ)/(3 η^2) ((1 - (2 χ)/η + 1/(1 - 2 χ / η)) BesselK[2 / 3, (4 χ)/(3 η^2 (1 - 2 χ / η))] -
      Integrate[BesselK[1 / 3, t], {t, (4 χ)/(3 η^2 (1 - 2 χ / η)), ∞}]),
   {χ, 0, 1}] // Normal, {χ > 0, η > 0}] // FullSimplify
```

Out[29]=

$$\begin{cases} -\frac{4\pi\chi}{3\sqrt{3}\,\eta^2} + \frac{2\left(\frac{2}{3}\right)^{1/3}\chi^{1/3}\,\mathrm{Gamma}\left[\frac{2}{3}\right]}{\eta^{2/3}} & \eta\ (\eta - 2\chi)\ \chi > 0\ \&\&\ \eta \geq 2\chi \\ -\frac{4\,(-1)^{2/3}\,\pi\,(-\eta)^{2/3}\,\chi}{3\sqrt{3}\,\eta^{8/3}} + \frac{2\left(\frac{2}{3}\right)^{1/3}\chi^{1/3}\,\mathrm{Gamma}\left[\frac{2}{3}\right]}{\eta^{2/3}} & \mathrm{True} \end{cases}$$

Figure B1

In[30]:=
```
W[x_] := ProductLog[0, x] (* Lambert Function *)

w0 = 2; (* [μm] *)
λ = 0.8; (* [μm] *)
k = 2 π / λ; (* [1/μm] *)
ω = 2 π / λ; (* [1/μm] *)
f = 9; (* [μm] *)
Int = 10^22; (* W/cm^2*)
(* Compton time *)
τC = 10^-6.5;(* ℏ/(m c^2)*) (* ??? *)
ESch = 1.3*10^16; (* [V/cm] *)
E0 = Sqrt[(2 Int)/(c ϵ0)] // N;(* [V/cm]*)

ξ[θ_] := Sqrt[Sin[θ]^2/((1 + Cos[θ])^2 k^2 w0^2) + (2 Log[2])/(k^2 f^2)] (*(B.6)*)
dτdϕ[ϕ_, θ_] := -(5 α)/(2 × √3 ω τC) E0/ESch Abs[Sin[ϕ]] Exp[-ξ[θ]^2 ϕ^2] (*(B.5)*) (**)
τf[ϕ1_, θ_] := (5 α)/(2 × √(3 π) ω τC) E0/ESch (1 - Erf[ξ[θ] ϕ1])/ξ[θ] (*(B.7)*)
fϕ1[ϕ1_, θ_] := Exp[-τf[ϕ1, θ]] Abs[dτdϕ[ϕ1, θ]] (*(B.8)*)
ϕ1mp[θ_] := Sqrt[1/(2 ξ[θ]^2) W[1/ξ[θ]^2 ((5 α)/(√6 π) 1/(ω τC) E0/ESch)^2]] (* (B.9) *)

plt = Plot[{fϕ1[ϕ1 , π / 3], fϕ1[ϕ1 , π / 6]}, {ϕ1, 10, 100},
   PlotRange → {{10, 90}, {0, 0.085}}, Axes → False, Frame → True,
   PlotStyle → {Blue, Red}, Filling → 0, FrameLabel → {"ϕ1", "f(ϕ1;θ)"}];
line1 = Graphics[{Blue, Line[{{ϕ1mp[π / 3], 0}, {ϕ1mp[π / 3], 0.09}}]}];
line2 = Graphics[{Red, Line[{{ϕ1mp[π / 6], 0}, {ϕ1mp[π / 6], 0.09}}]}];
Show[{plt, line1, line2}]
```

Out[48]=